

Dieser Datenabgleich ist für folgende Arbeitsweise im Verein gedacht:

Die Daten werden generell nur auf einen Arbeitsplatz bearbeitet. Diese Daten werden dann auf einen oder mehrere andere Arbeitsplätze übertragen, wobei gleichzeitig ein Update der KGV Software erfolgt. Auf diesen zusätzlichen Arbeitsplätzen können dann Recherchen durchgeführt werden oder aber bei Bedarf z.B. Arbeitsstunden bearbeitet werden. Die dann per XML Export/Import auf den Hauptrechner übertragen werden. *Dazu sollte es aber eine zusätzliche kurze Einweisung geben.*

Als Hilfsmittel wird hier das Programm 7-Zip genutzt welches von dem Link: <https://www.7-zip.de/> heruntergeladen werden kann. Das Programm ist in 32 und 64 Bit verfügbar. Mit Hilfe von 7-Zip werden die Daten in ein Archiv gepackt welches verschlüsselt werden kann. Dieses Archiv kann dann über eine Cloud z.B. OneDrive oder per Stick übertragen werden.

Gesteuert wird das über zwei CMD Dateien (KGV-Export.cmd und KGV-Import.cmd) die entsprechend angepasst werden müssen und dann über ein Icon gestartet werden. Beim Anpassen der CMD Dateien inklusive Einweisung würde ich bei Bedarf helfen, denn den meisten wird das zu kompliziert sein. Aber einmal eingerichtet ist die Handhabung auch für den Laien sehr einfach.

Diese CMD Dateien können an beliebigen Stellen abgelegt werden, ich empfehle im Programm-Ordner der KGV Software einen Ordner 7Z anzulegen und da die CMD Datei und die Hilfsdatei abzulegen. Als Arbeitsordner kann der Temp-Ordner der KGV-Software genutzt werden.

In den nachfolgenden Beispielen ist die Datenübertragung über OneDrive und bei Standardinstallation in C:\KGV dargestellt.

Was ist zu beachten:

Zweimal Doppelpunkt „:“ Kennzeichnet eine Kommentarzeile.

Bei den Variablen, die mit „set“ gesetzt werden, ist zu beachten, dass zwischen Variable Name Gleichheitszeichen und Wert kein Leerzeichen sein darf. Der Wert selbst kann Leerzeichen enthalten.

```
set test=Test Variable      richtig
set test = Test Variable    falsch
```

In der CMD werden die Variablen mit den Namen der zwischen zwei % Zeichen gesetzt wird aufgerufen: %test% würde hier „Test Variable“ ergeben.

Beispiel der Export Datei mit OneDrive als Cloud und Standardordner C:\KGV

```
@echo off
::Setzen der Variable für den Programmstart von 7-Zip
set Prg=C:\Programme\7-Zip\7z

::Setzen der Variable mit dem Installationsordner der KGV-Software
set Quelle=C:\KGV

::Setzen der Variable für den Ziel Ordner für 7-ZIP
set Ziel=C:\KGV\Temp\kgv.7z

::Setzen der Variable für das Copy Ziel nach erfolgreicher Archivierung
set CZiel1=C:\Users\test\OneDrive\KGV

::Setzen der Variable für die Datei in der die von der Archivierung
::ausgeschlossenen Ordner und Dateien stehen.
set Exclude=C:\kgv\7z\ex.txt

::Setzen der Variable für das Archiv Passwort. Das Passwort ist hier TestPSW.
:: Wird diese Variable nicht gesetzt, erfolgt keine Verschlüsselung!!!
Set PSW=-pTestPSW

::Löschen der alten Archivdatei
del %Ziel%

::Ausführen des Archivieren
%Prg% a -r -t7z -x@"%Exclude%" "%Ziel%" "%Quelle%\*.*" %PSW%

::Überprüfen ob die Archivierung erfolgreich war.
::Wenn ja Sprung zu E1 und kopieren des Archivs in die Cloud.
IF NOT errorlevel 1 goto E1
COLOR FC
echo.
@echo Fehler beim Archivieren, Datenbank eventuell geöffnet.
@echo Wenn KGV-Software aktiv bitte diese beenden und den
@echo Export erneut starten.
echo.
pause
exit

:E1

::Kopieren des Archivs in die Cloud
copy %Ziel% %CZiel1%

COLOR A
echo Sicherung erfolgreich
pause
```

Beispiel der Exclude Datei

```
DB_Backups
Arch*
xml
KGV_DB.INI
VEkgvgeneric.ini
TEMP
7Z
```

In der Exclude Datei werden alle Ordner und Dateien eingetragen, die nicht mit archiviert werden sollen. Die Einträge beziehen sich immer auf den root Ordner (Hauptordners) des Archivs. Ist ein Eintrag ungültig wird er übergangen. Wildcards sind möglich.

In dem Beispiel werden
die Ordner DB_Backups, xml, TEMP, 7z
alle Archiv Ordner (Arch*)
und die Dateien KGV_DB.INI und VEkgvgeneric.ini
von der Archivierung ausgeschlossen.

Damit wird ein unnötiges Aufblähen des Archivs verhindert und bei unterschiedlichen Installationen bleiben die Einstellungen im Ziel erhalten. Da ist in der Regel kein Autobackup notwendig, es sind damit auch unterschiedliche Installations-Laufwerke und Ordner möglich. Ebenfalls sind die Archive und Backups des Hauptrechners auf dem Ziel uninteressant.

Beispiel der Import Datei mit OneDrive als Cloud und Standardordner C:\KGV

```
@echo off
::Setzen der Variable für den Programmstart von 7-Zip
set Prg=C:\Programme\7-Zip\7z

::Setzen der Variable mit dem Installationsordner der KGV-Software
set Ziel=C:\KGV

::Setzen der Variable für den Quell Ordner für 7ZIP
set Quelle=C:\KGV\Temp\kgv.7z

::Setzen der Variable für den Quell Ordner in der Cloud
set CQuelle=C:\Users\Test\OneDrive\KGV\kgv.7z

::Setzen der Variable für das Archiv Passwort. Das Passwort ist hier TestPSW.
:: Wird diese Variable nicht gesetzt, erfolgt keine Verschlüsselung !!!
Set PSW=-pTestPSW

::Prüfen ob eine Archivdatei in der Cloud vorhanden ist.
::wenn ja Kopieren der Archivdatei in die Quell Datei für 7-Zip, sonst
::Sprung direkt zur Ausführung von 7-Zip
if not exist "%CQuelle%" goto NQ
Copy "%CQuelle%" "%Quelle%"

::Löschen der Archivdatei in der Cloud
del "%CQuelle%"

:NQ

::Aufführen des Imports/Extrahieren aus der Archivdatei.
%Prg% x -t7z "%Quelle%" %PSW% -o"%Ziel%" -y

::Abfrage ob Import Fehlerfrei war
IF NOT errorlevel 1 goto E1
COLOR 4
echo.
@echo Fehler beim Import, KGV Software eventuell aktiv.
@echo Bitte beenden und nochmals Probieren.
echo.
pause
exit

:E1

COLOR A
echo Import erfolgreich

::Wenn die KGV Software nach erfolgreichem Import gestartet werden soll muss
::die nächste Zeile ohne zweimal Doppelpunkt beginnen.
::start /b /d %Ziel% kgv.exe
```